# CONTENTS

# Mathematica Technical Supplement

## Loading the Mathematica Kernel

Performing a simple arithmetic operation, such as the one shown, can be used to activate the *Mathematica* Kernel, the software's computational engine.

```
1 + 1
```

```
2
```

## Function Definition

The function $f(t) = -2\,t\,e^{-t} + \frac{5\sin(2\,\pi t)}{t^2+1} + 1$ is defined.

```
f[t_] := 1 - 2 * t * Exp[-t] + 5 * Sin[2 * Pi * t] / (1 + t^2);
```

The use of ":=" (rather than simply "=") indicates a "delayed assignment". Using the colon tells *Mathematica* to return to the original definition each time you use the function.

Function evaluation ($f(t)$ is evaluated at $t = 3$):

```
f[3]
```

$$1 - \frac{6}{e^3}$$

```
N[f[3]]
```

```
0.701278
```

## Differentiation

The function $t^3 + t\sin t$ is differentiated.

```
D[t * Sin[t] + t ^ 3, t]
```

$$3\,t^2 + t\,\text{Cos}[t] + \text{Sin}[t]$$

The function, $f(t) = -2\,t\,e^{-t} + \frac{5\sin(2\,\pi t)}{t^2+1} + 1$, that was previously defined, is differentiated.

```
D[f[t], t]
```

$$-2\,e^{-t} + 2\,e^{-t}\,t + \frac{10\,\pi\,\text{Cos}[2\,\pi\,t]}{1+t^2} - \frac{10\,t\,\text{Sin}[2\,\pi\,t]}{(1+t^2)^2}$$

## Antidifferentiation

```
Integrate[3 t^2 + t Cos[t] + Sin[t], t]
```

```
t^3 + t Sin[t]
```

Note that the arbitrary constant is not appended.

## Graphing a Function

The function, $f(t) = -2\,t\,e^{-t} + \frac{5\sin(2\,\pi t)}{t^2+1} + 1$, that was previously defined, is plotted on the interval $-1 \le t \le 5$

```
Plot[f[t], {t, -1, 5}]
```




```
- Graphics -
```

Adjusting the range of ordinate values:

```
Plot[f[t], {t, -1, 5}, PlotRange → {-4, 8}]
```




```
- Graphics -
```

Superposing graphs:

```
Plot[{Sin[2*Pi*t], Exp[-t/2]}, {t, 0, 5}]
```




```
- Graphics -
```

Alternate approach; labels are given to the two individual plots and the **Show** command is then used to superpose them on the same graph.

```
p1 = Plot[Sin[2*Pi*t], {t, 0, 5}]
```




```
- Graphics -
```

```
p2 = Plot[Exp[-t/2], {t, 0, 5}]
```




```
- Graphics -
```

```
Show[p1, p2]
```




```
- Graphics -
```

# Piecewise-Defined Function

The function $\begin{cases} t^2, & 0 \le t \le 1 \\ e^{-(t-1)}, & 1 < t < \infty \end{cases}$ is defined and graphed.

```
f1[t_] := t^2 /; 0 ≤ t ≤ 1
```

```
f1[t_] := Exp[-(t - 1)] /; 1 < t
```

```
Plot[f1[t], {t, 0, 4}]
```




```
- Graphics -
```

# Root Finding

The intersection point abscissas on the graph shown represent solutions of the equation $\sin(2\pi t) - e^{-\frac{t}{2}} = 0$. The approximate values provided by the graph for two of these solutions are subsequently used as starting estimates in the FindRoot command.




Solutions of $\sin(2\pi t) - e^{-\frac{t}{2}} = 0$ near $t = 0.6$ and $t = 1.5$, respectively, are found.

```
FindRoot[Sin[2 * Pi * t] - Exp[-t / 2] == 0, {t, 0.6}]

{t → 0.340225}

FindRoot[Sin[2 * Pi * t] == Exp[-t / 2], {t, 1.5}]

{t → 1.41811}
```

# Direction Field

The appropriate graphics package is loaded and the direction field for the differential equation $y'(t) = 2\,y(t) + 1$ is plotted. The rectangular region $0 \leq t \leq 2$, $-1 \leq y \leq 1$ of the ty-plane is specified as the region of interest. The **PlotVectorField** command has more general applicability to autonomous two-dimensional systems. Note that in this special application, a 1 is placed in the first position and the right side of the first order differential equation appears in the second position ( that is, $\{1, 2 * y + 1\}$).

```
<< Graphics`PlotField`
```

```
PlotVectorField[{1, 2*y + 1}, {t, 0, 2}, {y, -1, 1}]
```



If coordinate axes are desired

```
PlotVectorField[{1, 2*y + 1}, {t, 0, 2}, {y, -1, 1}, Axes → True]
```




```
- Graphics -
```

## Symbolic Solution of First Order Linear Differential Equations

The general solution of the differential equation $y'(t) = 2\,y(t) + 1$ is determined. The fact that the equation is written as $y'(t) - 2\,y(t) = 1$ is not important; it could also have been written as $y'(t) - 2\,y(t) - 1 = 0$. Note the double equal sign, however.

```
DSolve[y'[t] - 2*y[t] == 1, y[t], t]
```

$$\left\{\left\{y[t] \to -\frac{1}{2} + e^{2t}\, C[1]\right\}\right\}$$

The initial value problem $y'(t) = 2\, y(t) + 1,\ y(0) = 2$ is solved.

```
DSolve[{y'[t] - 2 * y[t] == 1, y[0] == 2}, y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{2}\left(-1 + 5\, e^{2\,t}\right)\right\}\right\}$$

The prior output can be copied and pasted to create a function.

$$y[t\_] := \frac{1}{2}\left(-1 + 5\, e^{2\,t}\right)$$

# Limiting Behavior

The limiting behavior of three functions, $\frac{1}{2}\left(e^{2\,t} - 1\right)$, $e^{-t+\sin(t)}$ and $e^{\sin(t)}$, is examined as $t \to \infty$. The third function does not possess a limit. As the *Mathematica* output indicates, it oscillates in value over the interval $(e^{-1}, e^{1})$ as $t$ increases.

$$\text{Limit}\left[\frac{1}{2}\left(-1 + 5\, e^{2\,t}\right),\ t \to \text{Infinity}\right]$$

$\infty$

```
Limit[Exp[-t + Sin[t]], t → Infinity]
```

0

```
Limit[Exp[Sin[t]], t → Infinity]
```

$$\text{Interval}\left[\left\{\frac{1}{e},\ e\right\}\right]$$

# Symbolic Solution of Nonlinear First Order Differential Equations:

The initial value problem $y'(t) = y(t)^2,\ y(1) = 1$, is solved.

```
DSolve[{y'[t] - y[t]^2 == 0, y[1] == 1}, y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{2 - t}\right\}\right\}$$

The nonlinear differential equation $y'(t) = e^{-y(t)}$ is solved.

```
DSolve[y'[t] - Exp[-y[t]] == 0, y[t], t]
```

```
Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found.
```

```
{{y[t] → Log[t + C[1]]}}
```

The next example considers the nonlinear differential equation $\frac{d\,y(t)}{d\,t} - e^{-y(t)} + t^2 = 0$. Although it does not obviously fall into any of the categories discussed in the text, the change of dependent variable $u(t) = e^{y(t)}$ transforms it into a first order linear equation; therefore, we anticipate an explicit solution. *Mathematica* expresses this solution in terms of the natural logarithm and a special function called the Gamma function (see the Help Browser).

```
DSolve[y'[t] - Exp[-y[t]] + t^2 == 0, y[t], t]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found.

$$\left\{\left\{y[t] \to \frac{1}{3}\left(-t^3 + 3\,\mathrm{Log}\left[C[1] - \frac{t\,\mathrm{Gamma}\left[\frac{1}{3}, -\frac{t^3}{3}\right]}{3^{2/3}\,(-t^3)^{1/3}}\right]\right)\right\}\right\}$$

# Checking a Solution

Copy and paste the previous output to define the solution as a function.

$$y[t\_] := \frac{1}{3}\left(-t^3 + 3\,\mathrm{Log}\left[C[1] - \frac{t\,\mathrm{Gamma}\left[\frac{1}{3}, -\frac{t^3}{3}\right]}{3^{2/3}\,(-t^3)^{1/3}}\right]\right)$$

Form $y'(t) - e^{-y(t)} + t^2$.

```
D[y[t], t] - Exp[-y[t]] + t^2
```

$$-e^{\frac{1}{3}\left(t^3 - 3\,\mathrm{Log}\left[C[1] - \frac{t\,\mathrm{Gamma}\left[\frac{1}{3}, -\frac{t^3}{3}\right]}{3^{2/3}\,(-t^3)^{1/3}}\right]\right)} + t^2 + \frac{1}{3}\left(-3\,t^2 + \frac{3\left(e^{\frac{t^3}{3}} - \frac{t^3\,\mathrm{Gamma}\left[\frac{1}{3}, -\frac{t^3}{3}\right]}{3^{2/3}\,(-t^3)^{4/3}} - \frac{\mathrm{Gamma}\left[\frac{1}{3}, -\frac{t^3}{3}\right]}{3^{2/3}\,(-t^3)^{1/3}}\right)}{C[1] - \frac{t\,\mathrm{Gamma}\left[\frac{1}{3}, -\frac{t^3}{3}\right]}{3^{2/3}\,(-t^3)^{1/3}}}\right)$$

Simplify the output to see that it is in fact 0.

```
Simplify[%]
```

0

# Graphing Implicit Solutions

The nonlinear (separable) differential equation $y(t)\,y'(t) + t^3 = 0$ is solved.

```
DSolve[y[t] * y'[t] + t^3 == 0, y[t], t]
```

$$\left\{\left\{y[t] \to -\sqrt{2}\,\sqrt{-\frac{t^4}{4} + C[1]}\right\}, \left\{y[t] \to \sqrt{2}\,\sqrt{-\frac{t^4}{4} + C[1]}\right\}\right\}$$

The solutions lie on the level curves of the following function of two variables.

```
g[t_, y_] := (y^2) / 2 + (t^4) / 4
```

Some representative level curves of $g(t, y)$ are graphed.

```
ContourPlot[g[t, y], {t, -2, 2}, {y, -2, 2}, ContourShading → False,
 Axes → Automatic, AxesOrigin → {0, 0}, PlotPoints → 100]
```




```
- ContourGraphics -
```

## Symbolic Solution of Second Order Linear Differential Equations

The general solution of the second order linear nonhomogeneous differential equation $y''(t) - 4\,y(t) = e^{-2t}$ is obtained.

```
DSolve[y''[t] - 4 * y[t] == Exp[-2 * t], y[t], t]
```

$$\left\{\left\{y[t] \to -\frac{1}{16}\, e^{-2t}\,(1 + 4\,t) + e^{2t}\,C[1] + e^{-2t}\,C[2]\right\}\right\}$$

The initial value problem $y''(t) + 4\,y(t) = t + 1,\ y(0) = 0,\ y'(0) = 2$ is solved.

```
DSolve[{y''[t] + 4 * y[t] == t + 1, y[0] == 0, y'[0] == 2}, y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{8}\,(2 + 2\,t - 2\,\mathrm{Cos}[2\,t] + 7\,\mathrm{Sin}[2\,t])\right\}\right\}$$

## Higher Order Linear Differential Equations

The general solution of $\frac{d^3\,y(t)}{d\,t^3} + y(t) = t$ is obtained.

```
DSolve[y'''[t] + y[t] == t, y[t], t]
```

$$\left\{\left\{y[t] \to t + e^{-t}\,C[1] + e^{t/2}\,C[3]\,\mathrm{Cos}\left[\frac{\sqrt{3}\,t}{2}\right] + e^{t/2}\,C[2]\,\mathrm{Sin}\left[\frac{\sqrt{3}\,t}{2}\right]\right\}\right\}$$

The initial value problem $\frac{d^4 y(t)}{d t^4} - y(t) = 0,\ y(0) = 1,\ \frac{dy(0)}{d t} = 1,\ \frac{d^2 y(0)}{d t^2} = 0,\ \frac{d^3 y(0)}{d t^3} = -1$ is solved.

```
DSolve[{y''''[t] - y[t] == 0, y[0] == 1, y'[0] == 1, y''[0] == 0, y'''[0] == -1}, y[t], t]
```

$$\left\{\left\{y[t] \to \frac{1}{4} e^{-t} \left(1 + e^{2t} + 2 e^{t} \mathrm{Cos}[t] + 4 e^{t} \mathrm{Sin}[t]\right)\right\}\right\}$$

```
Simplify[%]
```

$$\left\{\left\{y[t] \to \frac{1}{4} \left(e^{-t} + e^{t} + 2 \mathrm{Cos}[t] + 4 \mathrm{Sin}[t]\right)\right\}\right\}$$

# Numerical Solution of an Initial Value Problem

The nonlinear pendulum differential equation is solved numerically using the **NDSolve** command.

```
sol = NDSolve[{y''[t] + Sin[y[t]] == 0, y[0] == 0, y'[0] == 2}, y[t], {t, 0, 8 * Pi}]
```

```
{{y[t] → InterpolatingFunction[{{0., 25.1327}}, <>][t]}}
```

The pendulum position is defined as a function. Notice the absence of the colon in the function definition.

```
ya[t_] = y[t] /. First[sol]
```

```
InterpolatingFunction[{{0., 25.1327}}, <>][t]
```

The pendulum velocity function is created.

```
yaprime[t_] = D[ya[t], t]
```

```
InterpolatingFunction[{{0., 25.1327}}, <>][t]
```

The pendulum position and velocity functions are plotted on two separate graphs.

```
Plot[ya[t], {t, 0, 8 * Pi}, PlotRange → {0, 10}]
```




```
- Graphics -
```

```
Plot[yaprime[t], {t, 0, 8*Pi}]
```




```
- Graphics -
```

# Matrices and First Order Systems

Defining a matrix.

```
A = {{1, -2, 0}, {-2, 1, 1}, {0, 1, 0 - 1}}
```

```
{{1, -2, 0}, {-2, 1, 1}, {0, 1, -1}}
```

Exhibiting the array.

```
MatrixForm[A]
```

$$\begin{pmatrix} 1 & -2 & 0 \\ -2 & 1 & 1 \\ 0 & 1 & -1 \end{pmatrix}$$

Finding eigenpairs.

```
Eigenvalues[A]
```

$$\Big\{\frac{1}{3} + \frac{19}{3\,(55 + 3\,i\,\sqrt{426})^{1/3}} + \frac{1}{3}\,(55 + 3\,i\,\sqrt{426})^{1/3},$$
$$\frac{1}{3} - \frac{19\,(1 + i\,\sqrt{3})}{6\,(55 + 3\,i\,\sqrt{426})^{1/3}} - \frac{1}{6}\,(1 - i\,\sqrt{3})\,(55 + 3\,i\,\sqrt{426})^{1/3},$$
$$\frac{1}{3} - \frac{19\,(1 - i\,\sqrt{3})}{6\,(55 + 3\,i\,\sqrt{426})^{1/3}} - \frac{1}{6}\,(1 + i\,\sqrt{3})\,(55 + 3\,i\,\sqrt{426})^{1/3}\Big\}$$

The matrix A is Hermitian and thus has real eigenvalues. This can be exhibited as follows.

```
N[ComplexExpand[%]]
```

```
{3.12489, -1.76156, -0.363328}
```

The three numbers listed above are the three real eigenvalues. Corresponding eigenvectors are now determined.

```
Eigenvectors[A]
```

$$\left\{\left\{-\frac{1}{6\,(55+3\,i\sqrt{426})^{2/3}}\left(157+2\,i\sqrt{426}+31\,(55+3\,i\sqrt{426})^{1/3}+i\sqrt{426}\,(55+3\,i\sqrt{426})^{1/3}+7\,(55+3\,i\sqrt{426})^{2/3}\right),\right.\right.$$
$$\left.\frac{4}{3}+\frac{19}{3\,(55+3\,i\sqrt{426})^{1/3}}+\frac{1}{3}\,(55+3\,i\sqrt{426})^{1/3},\,1\right\},$$
$$\left\{\frac{1}{2}+\frac{1}{2}\left(-\frac{2}{3}-\frac{19\,(1+i\sqrt{3})}{6\,(55+3\,i\sqrt{426})^{1/3}}-\frac{1}{6}\,(1-i\sqrt{3})\,(55+3\,i\sqrt{426})^{1/3}\right)\right.$$
$$\left(-\frac{4}{3}+\frac{19\,(1+i\sqrt{3})}{6\,(55+3\,i\sqrt{426})^{1/3}}+\frac{1}{6}\,(1-i\sqrt{3})\,(55+3\,i\sqrt{426})^{1/3}\right),$$
$$\left.\frac{4}{3}-\frac{19\,(1+i\sqrt{3})}{6\,(55+3\,i\sqrt{426})^{1/3}}-\frac{1}{6}\,(1-i\sqrt{3})\,(55+3\,i\sqrt{426})^{1/3},\,1\right\},$$
$$\left\{\frac{1}{2}+\frac{1}{2}\left(-\frac{2}{3}-\frac{19\,(1-i\sqrt{3})}{6\,(55+3\,i\sqrt{426})^{1/3}}-\frac{1}{6}\,(1+i\sqrt{3})\,(55+3\,i\sqrt{426})^{1/3}\right)\right.$$
$$\left(-\frac{4}{3}+\frac{19\,(1-i\sqrt{3})}{6\,(55+3\,i\sqrt{426})^{1/3}}+\frac{1}{6}\,(1+i\sqrt{3})\,(55+3\,i\sqrt{426})^{1/3}\right),$$
$$\left.\left.\frac{4}{3}-\frac{19\,(1-i\sqrt{3})}{6\,(55+3\,i\sqrt{426})^{1/3}}-\frac{1}{6}\,(1+i\sqrt{3})\,(55+3\,i\sqrt{426})^{1/3},\,1\right\}\right\}$$

The eigenvectors actually have real components. This can be seen as follows. (The $10^{-17}$ imaginary term arises due to numerical imprecision and should be ignored.)

```
N[ComplexExpand[%]]
```

$\{\{-3.88245 - 1.38778\times10^{-17}\,i,\ 4.12489,\ 1.\},$
$\{-0.551542,\ -0.761557,\ 1.\},\ \{0.933996,\ 0.636672,\ 1.\}\}$

The eigenvector corresponding to eigenvalue 3.1248854197645746 is formed as $x_1$ and the matrix product A$x_1$ is formed as a check.

```
x1 = {{-3.8824544433286095}, {4.124885419764574}, {1}}
```

```
{{-3.88245}, {4.12489}, {1}}
```

Check.

```
MatrixForm[x1]
```

$$\begin{pmatrix} -3.88245 \\ 4.12489 \\ 1 \end{pmatrix}$$

```
MatrixForm[A.x1]
```

$$\begin{pmatrix} -12.1322 \\ 12.8898 \\ 3.12489 \end{pmatrix}$$

## Eigenpair Calculation

The eigenpairs (that is eigenvalues and corresponding eigenvectors) of a matrix are simultaneously determined using the Eigensystem command. The matrix J chosen as an illustration has repeated eigenvalue $\lambda_1 = \lambda_2 = 2$ but only one linearly independent eigenvector $x_1 = \begin{pmatrix} 1 \\ 0 \end{pmatrix}$. *Mathematica* deals with this situation by incorrectly stating that $\begin{pmatrix} 0 \\ 0 \end{pmatrix}$ is the second eigenvector. This example illustrates an important point. No computer package is perfect and caution must always be used in accepting output of such packages as correct.

```
J = {{2, 1}, {0, 2}}

{{2, 1}, {0, 2}}

MatrixForm[J]
```

$$\begin{pmatrix} 2 & 1 \\ 0 & 2 \end{pmatrix}$$

```
Eigensystem[J]

{{2, 2}, {{1, 0}, {0, 0}}}
```

## Symbolic Solution of First Order Systems

The general solution of the linear first order system $\begin{pmatrix} y_1 \\ y_2 \end{pmatrix}' = \begin{pmatrix} 2 & -1 \\ -1 & 2 \end{pmatrix}\begin{pmatrix} y_1 \\ y_2 \end{pmatrix} + \begin{pmatrix} t \\ -1 \end{pmatrix}$ is determined.

```
DSolve[{y1'[t] == 2 * y1[t] - y2[t] + t, y2'[t] == -y1[t] + 2 * y2[t] - 1}, {y1[t], y2[t]}, t]
```

$$\Big\{\Big\{y1[t] \to \frac{1}{2} e^{t} (1 + e^{2t}) \left(e^{-3t}\left(-\frac{2}{9} - \frac{t}{6}\right) - \frac{e^{-t} t}{2}\right) - \frac{1}{2} e^{t} (-1 + e^{2t}) \left(e^{-3t}\left(\frac{2}{9} + \frac{t}{6}\right) - \frac{e^{-t} t}{2}\right) + \frac{1}{2} e^{t} (1 + e^{2t})\, C[1] - \frac{1}{2} e^{t} (-1 + e^{2t})\, C[2],$$

$$y2[t] \to -\frac{1}{2} e^{t} (-1 + e^{2t}) \left(e^{-3t}\left(-\frac{2}{9} - \frac{t}{6}\right) - \frac{e^{-t} t}{2}\right) + \frac{1}{2} e^{t} (1 + e^{2t}) \left(e^{-3t}\left(\frac{2}{9} + \frac{t}{6}\right) - \frac{e^{-t} t}{2}\right) - \frac{1}{2} e^{t} (-1 + e^{2t})\, C[1] + \frac{1}{2} e^{t} (1 + e^{2t})\, C[2]\Big\}\Big\}$$

```
Simplify[%]
```

$\left\{\left\{y1[t] \to \frac{1}{18}\left(-4 - 12\,t + 9\,e^{3\,t}\,(C[1] - C[2]) + 9\,e^{t}\,(C[1] + C[2])\right),\right.\right.$
$\left.\left. y2[t] \to \frac{1}{18}\left(4 - 6\,t - 9\,e^{3\,t}\,(C[1] - C[2]) + 9\,e^{t}\,(C[1] + C[2])\right)\right\}\right\}$

Note that the solution has the form $\begin{pmatrix} y_1 \\ y_2 \end{pmatrix} = \begin{pmatrix} \frac{-4-12\,t}{18} \\ \frac{4-6\,t}{18} \end{pmatrix} + \begin{pmatrix} \frac{9\,e^{3\,t}}{18} & \frac{9\,e^{t}}{18} \\ \frac{-9\,e^{3\,t}}{18} & \frac{9\,e^{t}}{18} \end{pmatrix} \begin{pmatrix} C_1 - C_2 \\ C_1 + C_2 \end{pmatrix}$.

The initial value problem $\begin{pmatrix} y_1 \\ y_2 \end{pmatrix}' = \begin{pmatrix} 2 & -1 \\ -1 & 2 \end{pmatrix} \begin{pmatrix} y_1 \\ y_2 \end{pmatrix} + \begin{pmatrix} t \\ -1 \end{pmatrix}$, $\begin{pmatrix} y_1(0) \\ y_2(0) \end{pmatrix} = \begin{pmatrix} 1 \\ 0 \end{pmatrix}$ is solved.

```
DSolve[{y1'[t] == 2*y1[t] - y2[t] + t,
  y2'[t] == -y1[t] + 2*y2[t] - 1, y1[0] == 1, y2[0] == 0}, {y1[t], y2[t]}, t]
```

$\left\{\left\{y1[t] \to \frac{1}{18}\left(-4 + 9\,e^{t} + 13\,e^{3\,t} - 12\,t\right),\ y2[t] \to \frac{1}{18}\left(4 + 9\,e^{t} - 13\,e^{3\,t} - 6\,t\right)\right\}\right\}$

# Plotting a Phase Plane Trajectory

The solution of the initial value problem $\begin{pmatrix} y_1 \\ y_2 \end{pmatrix}' = \begin{pmatrix} -1 & 3 \\ -3 & -1 \end{pmatrix} \begin{pmatrix} y_1 \\ y_2 \end{pmatrix}$, $\begin{pmatrix} y_1(0) \\ y_2(0) \end{pmatrix} = \begin{pmatrix} 1 \\ 0 \end{pmatrix}$ is determined. A parametric plot of the solution in the $(y_1, y_2)$-plane, for $0 \le t \le 5$, is created.

```
DSolve[{y1'[t] == -y1[t] + 3*y2[t],
  y2'[t] == -3*y1[t] - y2[t], y1[0] == 1, y2[0] == 0}, {y1[t], y2[t]}, t]
```

$\{\{y1[t] \to e^{-t}\,\mathrm{Cos}[3\,t],\ y2[t] \to -e^{-t}\,\mathrm{Sin}[3\,t]\}\}$

```
ParametricPlot[{ⅇ^-t Cos[3 t], -ⅇ^-t Sin[3 t]},
 {t, 0, 5}, PlotRange → {{-0.5, 1.2}, {-0.8, 0.4}}]
```




- Graphics -

## Phase Plane Analysis of Autonomous 2-Dimensional Systems

The vector field for the two-dimensional autonomous system

$$x'(t) = 5\,x(t) - x(t)^2 - x(t)\,y(t)$$
$$y(t)' = x(t)\,y(t) - 2\,y(t)$$

is plotted in the rectangle $-2 \le x \le 8,\ -6 \le y \le 6$. The initial **<< Graphics`PlotField`** command loads the graphics package.

```
<< Graphics`PlotField`
```

```
PlotVectorField[{5 * x - x^2 - x * y, x * y - 2 * y}, {x, -2, 8}, {y, -6, 6}, Axes → True]
```




- Graphics -

Phase plane behavior in the vicinity of the equilibrium point (0,0) is examined.

```
PlotVectorField[{5 * x - x^2 - x * y, x * y - 2 * y}, {x, -0.2, 0.2}, {y, -0.2, 0.2}]
```



- Graphics -

## Laplace Transforms

The Laplace transform of the function $h(t-1)+t\,e^{-t}+\cos(2\,t)$ is determined.

```
LaplaceTransform[UnitStep[t - 1] + t * Exp[-t] + Cos[2 * t], t, s]
```

$$\frac{e^{-s}}{s}+\frac{1}{(1+s)^2}+\frac{s}{4+s^2}$$

The Laplace transform of the 2×1 matrix-valued function $\begin{pmatrix} t^2 \\ 1+\sin(t) \end{pmatrix}$ is determined and subsequently displayed as a 2×1 matrix.

```
LaplaceTransform[{{t^2}, {1 + Sin[t]}}, t, s]
```

$$\left\{\left\{\frac{2}{s^3}\right\},\left\{\frac{1}{s}+\frac{1}{1+s^2}\right\}\right\}$$

```
MatrixForm[%]
```

$$\begin{pmatrix} \frac{2}{s^3} \\ \frac{1}{s}+\frac{1}{1+s^2} \end{pmatrix}$$

The inverse Laplace transform for each of the two previous cases is computed. Note that the outputs of the **LaplaceTransform** computations were simply copied and pasted into the **InverseLaplaceTransform** commands.

$$\texttt{InverseLaplaceTransform}\left[\frac{e^{-s}}{s}+\frac{1}{(1+s)^2}+\frac{s}{4+s^2},\ s,\ t\right]$$

$$e^{-t}\,t+\mathrm{Cos}[2\,t]+\mathrm{UnitStep}[-1+t]$$

$$\texttt{MatrixForm}\left[\texttt{InverseLaplaceTransform}\left[\left\{\left\{\frac{2}{s^3}\right\},\left\{\frac{1}{s}+\frac{1}{1+s^2}\right\}\right\},\ s,\ t\right]\right]$$

$$\begin{pmatrix} t^2 \\ 1+\mathrm{Sin}[t] \end{pmatrix}$$

## Graphing the Periodic Thrust Velocity

In section 7.4, solving the problem of a projectile propelled by a periodic thrust using Laplace transforms led to a velocity function of the form

$v(t)=e^{-t}+\sum_{n=0}^{\infty}(-1)^n\,(1-e^{-(t-n)}\,h(t-n))$, where $h(t)$ is the Heaviside (or unit step) function. We can define and graph this function in the following manner.

```
G[t_, n_] := ((-1)^n) * (1 - Exp[-(t - n)]) * UnitStep[t - n]

v[t_] := Exp[-t] + Sum[G[t, n], {n, 0, Infinity}]
```

```
Plot[v[t], {t, 0, 20}, PlotRange -> {0, 1.1}]
```



- Graphics -

## Series Expansions

The Series command can be used to compute  partial sums of Maclaurin and Taylor series expansions. As examples, consider the Maclaurin series expansion of Tan(ax), where a is a constant, and the Taylor series expansion of Cos(x) about $\frac{\pi}{4}$. Note that since Tan(ax) is an odd function, only odd powers appear in the Maclaurin expansion. The O( ) symbol appearing at the end indicates the order or "size" of the error incurred due to the series truncation.

```
Series[Tan[a * x], {x, 0, 8}]
```

$$a x + \frac{a^3 x^3}{3} + \frac{2 a^5 x^5}{15} + \frac{17 a^7 x^7}{315} + O[x]^9$$

```
Series[Cos[x], {x, Pi / 4, 6}]
```

$$\frac{1}{\sqrt{2}} - \frac{x - \frac{\pi}{4}}{\sqrt{2}} - \frac{(x - \frac{\pi}{4})^2}{2\sqrt{2}} + \frac{(x - \frac{\pi}{4})^3}{6\sqrt{2}} + \frac{(x - \frac{\pi}{4})^4}{24\sqrt{2}} - \frac{(x - \frac{\pi}{4})^5}{120\sqrt{2}} - \frac{(x - \frac{\pi}{4})^6}{720\sqrt{2}} + O\left[x - \frac{\pi}{4}\right]^7$$

## Equilibrium Points for Two-Dimensional Autonomous Systems

As an example consider the two-dimensional autonomous system:

$$x'(t) = x(t)\,y(t) - y(t) + x(t) - 1$$
$$y'(t) = x(t)\,y(t) - 2\,y(t)$$

The equilibrium points of this system, $(-1, 0)$ and $(2. - 1)$, can be obtained as follows.

```
Solve[{x * y - y + x - 1 == 0, x * y - 2 * y == 0}, {x, y}]
```

```
{{x → 1, y → 0}, {x → 2, y → -1}}
```

## Linearization About an Equilibrium Point

Suppose we want to determine the linearization of the autonomous system

$$x'(t) = x(t)\,y(t) - y(t) + x(t) - 1$$
$$y'(t) = x(t)\,y(t) - 2\,y(t)$$

about the equilibrium point $(2, -1)$. We first define two functions.

```
f[x_, y_] := x * y - y + x - 1;
```

```
g[x_, y_] := x * y - 2 * y;
```

The matrix of partial derivatives is constructed and displayed as an array.

```
J[x_, y_] = MatrixForm[{{D[f[x, y], x], D[f[x, y], y]}, {D[g[x, y], x], D[g[x, y], y]}}]
```

$$\begin{pmatrix} 1 + y & -1 + x \\ y & -2 + x \end{pmatrix}$$

The matrix J is evaluated at $(2, -1)$ and displayed as an array.

```
J[2, -1]
```

$$\begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix}$$

# Partial Differential Equations and Fourier Series

## Defining and Graphing Functions of Two Variables

```
u[x_, t_] := Exp[-(x^2 + t) / 3] * (2 + Cos[x - 2 * t]);
```

```
Plot3D[u[x, t], {x, -5, 5}, {t, 0, 2}, PlotRange → {0, 3}, PlotPoints → 50]
```




```
- SurfaceGraphics -
```

The **PlotPoints** option permits one to increase the number of sampling points, creating a smoother, less choppy-looking, surface.

## Polar Plots

The function $u(r, \theta) = \frac{1}{r} + r\sin(3\,\theta)/2$, defined on the annulus $1 \le r \le 3,\ 0 \le \theta < 2\pi$, is plotted. Recall that $x = r\cos(\theta),\ y = r\sin(\theta)$.

```
ParametricPlot3D[{r * Cos[theta], r * Sin[theta], 1 / r + r * Sin[3 * theta] / 2},
 {r, 1, 3}, {theta, 0, 2 * Pi}, PlotPoints → 80]
```




```
- Graphics3D -
```

## Fourier Series

Consider the problem of computing the Fourier series for the periodic extension of the function

$$f(x) = \begin{cases} x, & 0 \le x \le 1 \\ -1, & 1 < x < 3 \end{cases}.$$

Once the series is formed we plot representative partial sums.

Fourier series coefficients (For this example, $l = 3/2$).

```
a[n_] := (2 / 3) * (Integrate[x * Cos[n * Pi * x / (3 / 2)], {x, 0, 1}] +
    Integrate[(-1) * Cos[n * Pi * x / (3 / 2)], {x, 1, 3}])

a[n]
```

$$\frac{2}{3}\left(-\frac{9}{4\,n^2\,\pi^2}+\frac{9\,\mathrm{Cos}\left[\frac{2\,n\,\pi}{3}\right]}{4\,n^2\,\pi^2}+\frac{3\,\mathrm{Sin}\left[\frac{2\,n\,\pi}{3}\right]}{n\,\pi}-\frac{3\,\mathrm{Sin}[2\,n\,\pi]}{2\,n\,\pi}\right)$$

```
a[0]

-1
```

$$\mathrm{aa}[n\_] := \frac{2}{3}\left(-\frac{9}{4\,n^2\,\pi^2}+\frac{9\,\mathrm{Cos}\left[\frac{2\,n\,\pi}{3}\right]}{4\,n^2\,\pi^2}+\frac{3\,\mathrm{Sin}\left[\frac{2\,n\,\pi}{3}\right]}{n\,\pi}-\frac{3\,\mathrm{Sin}[2\,n\,\pi]}{2\,n\,\pi}\right);$$

```
b[n_] := (2 / 3) * (Integrate[x * Sin[n * Pi * x / (3 / 2)], {x, 0, 1}] +
    Integrate[(-1) * Sin[n * Pi * x / (3 / 2)], {x, 1, 3}])
```

```
b[n]
```

$$\frac{2}{3}\left(-\frac{3\,\text{Cos}[\frac{2\,n\,\pi}{3}]}{n\,\pi}+\frac{3\,\text{Cos}[2\,n\,\pi]}{2\,n\,\pi}+\frac{9\,\text{Sin}[\frac{2\,n\,\pi}{3}]}{4\,n^2\,\pi^2}\right)$$

$$\text{bb[n\_]} := \frac{2}{3}\left(-\frac{3\,\text{Cos}[\frac{2\,n\,\pi}{3}]}{n\,\pi}+\frac{3\,\text{Cos}[2\,n\,\pi]}{2\,n\,\pi}+\frac{9\,\text{Sin}[\frac{2\,n\,\pi}{3}]}{4\,n^2\,\pi^2}\right)$$

Partial sum function $S(x, N) = \frac{a_0}{2} \sum_{n=1}^{N} [a_n \cos(\frac{n\pi x}{3/2}) + b_n \sin(\frac{n\pi x}{3/2})]$ is defined and plotted for a representative value of $N$.

```
S[x_, N_] := -1 / 2 + Sum[aa[n] * Cos[n * Pi * x / (3 / 2)] + bb[n] * Sin[n * Pi * x / (3 / 2)], {n, 1, N}]

Plot[S[x, 20], {x, -3, 3}]
```




```
- Graphics -
```

**Alternate Approach:**

If a partial sum of a fixed number of terms is to be formed and graphed, the calculations can usually be done more quickly by first forming numerical arrays of the Fourier coefficients and then using these numerical values to form the partial sum. This approach is illustrated below for $N$=20.

```
dataA = N[Table[a[n], {n, 1, 20}]]

{0.323356, -0.332658, 0., 0.123584, -0.119385, 0.,
 0.0741088, -0.0724782, 0., 0.0528532, -0.0520049, 0., 0.041061,
 -0.0405438, 0., 0.0335675, -0.0332199, 0., 0.0283858, -0.0281364}

aa = {0.323356232226532`, -0.332657613509711`, 0.`,
   0.12358393240574425`, -0.11938468561216882`, 0.`, 0.0741087674440262`,
   -0.07247818479014993`, 0.`, 0.052853162910226605`, -0.05200488027136341`, 0.`,
   0.041060964362651095`, -0.040543761219899735`, 0.`, 0.033567537748255515`,
   -0.03321994424001981`, 0.`, 0.028385807063209943`, -0.028136376429077754`};
```

```
dataB = N[Table[b[n], {n, 1, 20}]]

{0.76824, 0.285405, -0.106103, 0.167381, 0.122059, -0.0530516, 0.0936318,
 0.0775209, -0.0353678, 0.0649782, 0.0567868, -0.0265258, 0.0497496,
 0.0448013, -0.0212207, 0.0403029, 0.0369928, -0.0176839, 0.0338709, 0.0315019}

bb = {0.7682398508312406`, 0.28540486656787584`,
   -0.1061032953945969`, 0.16738119799587403`, 0.12205915133496992`,
   -0.05305164769729845`, 0.09363180581707609`, 0.077520907819953`,
   -0.035367765131532294`, 0.06497817802139473`, 0.0567867555171879`,
   -0.026525823848649224`, 0.04974956875291253`, 0.044801309870828976`,
   -0.02122065907891938`, 0.0403028767044725`, 0.03699278910652327`,
   -0.017683882565766147`, 0.03387090236412106`, 0.03150193842221992`};

ss[x_] :=
 -1 / 2 + Sum[aa[[n]] * Cos[n * Pi * x / (3 / 2)] + bb[[n]] * Sin[n * Pi * x / (3 / 2)], {n, 1, 20}]
```

```
Plot[ss[x], {x, -3, 3}]
```




```
- Graphics -
```

## Multivariate Fourier Series

Consider the function $u(x, y)$, defined on the square $0 \le x \le 1,\ 0 \le y \le 1$,

$$u(x, y) = \begin{cases} 4, & \frac{1}{4} \le x \le \frac{3}{4},\ \frac{1}{3} \le y \le \frac{2}{3} \\ 0, & \text{otherwise} \end{cases}.$$

We will graph the function, compute the Fourier series coefficients of the two-dimensional odd periodic extension of this function and then plot a representative partial sum surface.

```
f2[x_] := 0 /; x < 1 / 4;

f2[x_] := 1 /; 1 / 4 ≤ x ≤ 3 / 4;

f2[x_] := 0 /; x > 3 / 4;

g2[y_] := 0 /; y ≤ 1 / 3;;

g2[y_] := 1 /; 1 / 3 ≤ y ≤ 2 / 3;

g2[y_] := 0 /; 2 / 3 < y;

u[x_, y_] := 4 * f2[x] * g2[y];
```

```
Plot3D[u[x, y], {x, 0, 1}, {y, 0, 1}, PlotPoints → 80]
```




- SurfaceGraphics -

The Fourier series of the two-dimensional odd periodic extension has the form $u(x, y) = \sum_{m,n=1}^{\infty} c_{\mathrm{mn}} \sin(\frac{m\pi x}{2})\sin(\frac{n\pi y}{2})$, where $c_{\mathrm{mn}} = 16\int_0^1\int_0^1 u(x, y)\sin(m\,\pi\,x)\sin(n\,\pi y)\,d\,x\,d\,y$. (Note that, because of the special form of u(x,y), the coefficient evaluation reduces to $c_{\mathrm{mn}} = 16\int_{\frac{1}{4}}^{\frac{3}{4}} \sin(m\,\pi\,x)\,d\,x \int_{\frac{1}{3}}^{\frac{2}{3}} \sin(n\,\pi\,y)\,d\,y$ .)

```
c[m_, n_] := 16 * Integrate[Sin[m * Pi * x] * Sin[n * Pi * y], {x, 1 / 4, 3 / 4}, {y, 1 / 3, 2 / 3}];

S[x_, y_, M_, N_] := Sum[c[m, n] * Sin[m * Pi * x] * Sin[n * Pi * y], {m, 1, M}, {n, 1, N}];

Plot3D[S[x, y, 20, 20], {x, 0, 1}, {y, 0, 1}, PlotPoints → 50]
```




- SurfaceGraphics -

## Symbolic Solution of a Two-Point Boundary Value Problem

We solve and graph the solution of the two-point boundary value problem $y(x)'' + y(x)' + y(x) = 0,\ y(0) = 0,\ y(1)' = 1$.

```
DSolve[{y''[x] + y'[x] + y[x] == 0, y[0] == 0, y'[1] == 1}, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{2\, e^{\frac{1}{2}-\frac{x}{2}}\, \text{Sin}\left[\frac{\sqrt{3}\, x}{2}\right]}{\sqrt{3}\, \text{Cos}\left[\frac{\sqrt{3}}{2}\right] - \text{Sin}\left[\frac{\sqrt{3}}{2}\right]}\right\}\right\}$$

$$\text{Plot}\left[\frac{2\, e^{\frac{1}{2}-\frac{x}{2}}\, \text{Sin}\left[\frac{\sqrt{3}\, x}{2}\right]}{\sqrt{3}\, \text{Cos}\left[\frac{\sqrt{3}}{2}\right] - \text{Sin}\left[\frac{\sqrt{3}}{2}\right]}, \{x, 0, 1\}\right]$$




- Graphics -

## Numerical Solution of a Two-Point Boundary Value Problem

The two-point boundary value problem $r(t)'' - r(t) = 0,\ r(0) = 0,\ r(1)' = 1$ is solved numerically using the **NDSolve** command. Functions for the position and velocity, labelled as ra(t) and raprime(t), are defined and plotted on the same graph. Note the absence of the colon in the definition of these functions.

```
rsol = NDSolve[{r''[t] - r[t] == 0, r[0] == 0, r'[1] == 1}, r[t], {t, 0, 1}];

ra[t_] = r[t] /. First[rsol]

InterpolatingFunction[{{0., 1.}}, <>][t]

raprime[t_] = D[ra[t], t];
```

```
Plot[{ra[t], raprime[t]}, {t, 0, 1}]
```




- Graphics -

## Symbolic Solution of a Two-Point Boundary Value Problem Involving a Linear First-Order System

The two-point boundary value problem

$$\begin{pmatrix} I_1 \\ I_2 \end{pmatrix}' = \begin{pmatrix} 1 & -1 \\ 1 & -1 \end{pmatrix} \begin{pmatrix} I_1 \\ I_2 \end{pmatrix}, \quad \begin{matrix} I_1(0) = 1 \\ I_1(1) - \frac{1}{2} I_2(1) = 0 \end{matrix}.$$

is solved.

```
DSolve[{I1'[x] == I1[x] - I2[x], I2'[x] == I1[x] - I2[x],
  I1[0] == 1, I1[1] - (1/2) * I2[1] == 0}, {I1[x], I2[x]}, x]
```

$$\left\{\left\{I1[x] \to \frac{2-x}{2}, I2[x] \to \frac{3-x}{2}\right\}\right\}$$